NPS-71-89-001

NAVAL POSTGRADUATE SCHOOL

Monterey, California

SEARCH TIMES AND FALSE TARGETS

by

R.N. FORREST

JULY 1989

Prepared for:
Chief of Naval Operations
Washington D.C. 20350

NAVAL POSTGRADUATE SCHOOL
Monterey, California

Rear Admiral R. W. West, Jr.
Superintendent

Harrison Shull
Provost

This report was prepared in conjunction with research conducted for the Chief of Naval Operations and funded by the Naval Postgraduate School.

REPORT DOCUMENTATION PAGE

1a REPORT SECURITY CLASSIFICATION UNCLASSIFIED	1b RESTRICTIVE MARKINGS
2a SECURITY CLASSIFICATION AUTHORITY	3 DISTRIBUTION/AVAILABILITY OF REPORT Approved for public release; distribution is unlimited.
2b DECLASSIFICATION/DOWNGRADING SCHEDULE	
4 PERFORMING ORGANIZATION REPORT NUMBER(S) NPS-71-89-001	5 MONITORING ORGANIZATION REPORT NUMBER(S)
6a NAME OF PERFORMING ORGANIZATION Naval Postgraduate School 6b OFFICE SYMBOL (*If applicable*) 71	7a NAME OF MONITORING ORGANIZATION Naval Postgraduate School
6c ADDRESS (*City, State, and ZIP Code*) Monterey, CA 93943-5000	7b ADDRESS (*City, State, and ZIP Code*) Monterey, CA 93943-5000
8a NAME OF FUNDING/SPONSORING ORGANIZATION Chief of Naval Operations 8b OFFICE SYMBOL (*If applicable*) OP-71	9 PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER Direct Funding, O&MN
8c ADDRESS (*City, State, and ZIP Code*) Washington D.C. 20350	10 SOURCE OF FUNDING NUMBERS PROGRAM ELEMENT NO / PROJECT NO / TASK NO / WORK UNIT ACCESSION NO.

11 TITLE (*Include Security Classification*)
Search Times and False Targets

12 PERSONAL AUTHOR(S)
R.N. Forrest

13a TYPE OF REPORT	13b TIME COVERED	14 DATE OF REPORT (*Year, Month, Day*)	15 PAGE COUNT
Technical	FROM _____ TO _____	July 1989	30

16 SUPPLEMENTARY NOTATION

17 COSATI CODES			18 SUBJECT TERMS (*Continue on reverse if necessary and identify by block number*)
FIELD	GROUP	SUB-GROUP	Search, False targets, Search Time

19 ABSTRACT (*Continue on reverse if necessary and identify by block number*)

The report describes an analysis of some simulated search time data. The purpose of the report is to provide a guide for estimating the impact of false targets on the time a searcher is exposed to hostile action. The program that was used to generate the search time data is listed in the report.

20 DISTRIBUTION/AVAILABILITY OF ABSTRACT ☒ UNCLASSIFIED/UNLIMITED ☐ SAME AS RPT ☐ DTIC USERS	21 ABSTRACT SECURITY CLASSIFICATION UNCLASSIFIED
22a NAME OF RESPONSIBLE INDIVIDUAL R.N. FORREST	22b TELEPHONE (*Include Area Code*) 408-646-2653 22c OFFICE SYMBOL 55Fo

☆ U.S. Government Printing Office: 1986—606-243

TABLE OF CONTENTS

I. Introduction

The subject of this report is an analysis of some simulated search time data. The purpose of the report is to provide a guide for estimating the impact of false targets on the time a searcher is exposed to hostile action. The extent to which this purpose is achieved may be evident from the conjectures that are given in Section IV.

The analysis of the simulated search time data suggests that the median is a superior to the mean as a measure of the effect of false targets on search time. The data was generated by an event step Monte Carlo simulation. This approach was chosen because of the intractable character of the probabilistic structure of the search model on which the simulation is based. However, the search model is limited. In the model, the relative positions of an assembly of targets remain constant during a search and, with respect to the assembly of targets, the searcher's speed is constant. In addition, false targets are identical with respect to their detection and classification characteristics and both the detection and the classification capability of a searcher are described in terms of definite range laws. That is, a target is detected or classified when the range between the target and the searcher equals a specified value.

The simulation is implemented by a program named TARGET.BAS that is listed in Appendix 1.

II. The Search Model

Some characteristics of the simulation program and the search model on which it is based are discussed in this section. The program simulates a search on a plane surface by a searcher who knows that if a true target is present, it is within a region bounded by a circle. The location of the center and the radius of the bounding circle are known to the searcher. The searcher moves on a straight track until either a detection occurs or the searcher reaches the bounding circle. If a detection occurs, relative to the region, the searcher moves on a straight track toward the detected target, either true or false, until the classification range is reached. If the target is the true target, then the search terminates. Otherwise, the search continues until the true target is detected and classified or, if a true target is not present, until all of the false targets are classified. If more than one target is within the searcher's detection range, the searcher classifies the nearest target and then, if it is not the true target, the next nearest target and so on. Figure 7 in Appendix 4 shows a simulated search track and the geometry of a simulated search region.

The searcher's knowledge of the location of a target is determined by the detection range that is specified in the simulation. If the range is greater than the diameter of the circle that bounds the search region, then the position of all of the targets will be known to the searcher at the beginning of the search. When the searcher's detection range is less than the

diameter of the bounding circle, in general, not all of the target locations will be known and, to this extent, the search will have a random character that is determined by the random course choices of the searcher in addition to that determined by the location of the targets and the search starting point.

For each simulation trial, if there is a true target in the search area, the program determines the time to detect and classify it. Otherwise, the program determines the time to detect and classify the false targets that are within the search area. The program generates values for the following statistics: the average search duration and its standard deviation, the average number of classifications and its standard deviation. If the option to generate a search statistics data file is chosen, these values are stored in this file. If the option to generate a search duration histogram data file is also chosen, the number of entries in each histogram cell is also stored in this file. A program that is listed in Appendix 2 will read and print these files. The program is named TARGETPR.BAS.

III. Program Generated Search Times

Some data that were generated by the program named TARGET.BAS are presented in this section in Figures 1 through 5. These figures show estimates of the cumulative distribution functions of search times for various search conditions that are based on 50,000 trials. Two general conditions are the circular boundaries of the search regions and the cosine law that determines the reflectiom of search tracks at these boundaries. Two specific conditions are the equality between the true target and the false target detection ranges and the equality between the true target and the false target classification ranges.

The implication of the output data relative to the purpose of the analysis is discussed in Section IV.



Figure 1 Search time cumulative distribution functions for a 50 nautical mile radius, 0 false targets, a 5 knot search speed, a 5 knot classification speed, a 5 nautical mile classification range and detection ranges of 5, 10, 15, 30 and 200 nautical miles.



Figure 2 Search time cumulative distribution functions for a 50 nautical mile radius, 5 false targets, a 5 knot search speed, a 5 knot classification speed, a 5 nautical mile classification range and detection ranges of 5, 10, 15, 30 and 200 nautical miles.



Figure 3 Search time cumulative distribution functions for a 50 nautical mile radius, 10 false targets, a 5 knot search speed, a 5 knot classification speed, a 5 nautical mile classification range and detection ranges of 5, 10, 15, 30 and 200 nautical miles.



Figure 4 Search time cumulative distribution functions for a 50 nautical mile radius, 15 false targets, a 5 knot search speed, a 5 knot classification speed, a 5 nautical mile classification range and detection ranges of 5, 10, 15, 30 and 200 nautical miles.



Figure 5 Search time cumulative distribution functions for a 50 nautical mile radius, 20 false targets, a 5 knot search speed, a 5 knot classification speed, a 5 nautical mile classification range and detection ranges of 5, 10, 15, 30 and 200 nautical miles.

IV. Some Conjectures Based on the Simulation Program Output

In some cases of interest, the motion of a target relative to a searcher can be used as a classification clue and the range between a target and the searcher at which classification is achieved is a random quantity. The search model does not account for this except in an average sense. In effect, the search model describes a situation in which classification clues are aggregated so that classification occurs when the range between the searcher and a target has a specified value. TARGET.BAS, the program that was used to implement the search model, requires that this average classification range be the same for all false targets. In addition to the limitations in the description of the classification process, the search model describes a situation in which detection also occurs when the range between the searcher and a target has a specified value and TARGET.BAS also requires that this range be the same for all false targets.

The effect of the limitations of the search model on the utility of data that is generated with the program, in particular, the limitations imposed by the lack of relative motion and the requirement that the search region be bounded by a circle, as well as other limitations, can only be conjectured. The conjecture that was adopted in choosing the search model is that such data would be adequate for the purpose of the analysis.

Figure 6 which is based on the data presented in Section III indicates the effects of false targets on search times subject to the conditions of the model. These conditions include the

following: equal true target and false target detection ranges, equal true target and false target classification ranges, and a classification range of 5 nautical miles. Referring to Figure 6, note that for detection ranges of 30 nautical miles and 200 nautical miles, the median search time increases as the number of false targets increase. However, for detection ranges of 5 nautical miles, 10 nautical miles and 15 nautical miles, the median search time decreases as the number of false targets increase.

Given detection ranges that are approximately the same, for a search in a region bounded by a circle containing the true target and false targets that are distributed randomly over the region, the above results suggest the following conjecture: If the searcher's detection range for the targets is not greater than the radius of the bounding circle, then the exposure of the searcher will not be significantly increased by the presence of false targets.

For 0 false targets and detection ranges of 5 nautical miles, 10 nautical miles and 15 nautical miles, the median search times are 116 hours, 59 hours and 39 hours. For 5 false targets, and these detection ranges, the median search times are 103 hours, 51 hours and 33 hours. For a search region bounded by a circle with a radius of 50 nautical miles and a searcher with a speed of 5 knots, as shown in Appendix 3, for these detection ranges, the random search model gives a median search times of 109 hours, 54 hours and 36 hours. For 20 false targets, and

these detection ranges, the median search times are 89 hours, 46 hours and 32 hours.

Relative to the parameter values that were used to generate the data, these results suggest a second conjecture: For target densities of .001 targets per square nautical mile or less, the simulation search model is equivalent to the random search model. (For the 5 nautical mile detection range, one might expect the correspondence between the models to be the greatest since then classification occurs at detection in the simulation search model as it does in the random search model.)

Because of the charicteristics of the simulation search model, in particular the reflection law, the second conjecture is supported by the work of Lalley and Robbins that appears in Reference 1.

The second conjecture suggests that the random search model might be used in order to estimate an upper bound on the exposure associated with a systematic search when a searcher's detection range is of the order of one-tenth the diameter of the circle that just encompasses the search region.

When the searcher's detection range for both the true target and the false targets is greater than the diameter of the circle that just encompasses the search region, after the classification of a false target, the searcher moves toward the nearest unclassified target. Since the distance between targets cannot be greater than the diameter of the bounding circle, clearly, in this case, a loose upper bound on the search time is $2 \cdot r \cdot n$

where r is the radius of the bounding circle and n is the number of targets.

Search models are described in Reference 2 and Reference 3 that involve false targets under conditions related to conditions of the model that is described in this report. In addition, they and Reference 1 include a number of other references that describe search models that involve false targets.



Figure 6 Plots of median search time as a function of the number of false targets for five different detection ranges: 5, 10, 15, 30, and 200 nautical miles.

Appendix 1.

The program that is listed below was used to generate the data in Section III. The program is designed to be used with a monitor with a capability corresponding to BASIC screen mode 9. This condition can be changed by changing the mode number in Line 540. For example, for a monitor with CGA capability only, SCREEN 9 should be replaced by SCREEN 2. In addition, the program can be used without the DEFDBL A-Z specification in Line 20. To start a search at a random point on the search area bounding circle, input a value for the initial range from the center equal to the search area radius. Note that the initial range from the center must be less than or equal to the search area radius. Also note that the classification range must be less than or equal to the detection range. To generate a histogram of the search duration times, first, choose the search statistics data file option and then choose the search duration histogram file option. The standard output of the program includes the following estimates: search duration mean, search duration standard deviation, number of classifications mean and number of classifications standard deviation.

```
10 REM: TARGET.BAS
20 CLS : DEFDBL A-Z: N = 0: Q = 0: S = 0: U = 0: W = 0: DT = 0: DF = 0: KSUM = 0: PI = 4 * ATN(1): R$ =
"no": T$ = "no"
30 INPUT "number of trials"; M
40 INPUT "number of false targets"; N
50 DIM X(N + 1), Y(N + 1), R(N + 1), X1(N + 1), Y1(N + 1), D1(N + 1), D2(N + 1), D3(N + 1), H(7000)
60 IF N = 0 THEN B$ = "Y": GOTO 80
70 INPUT "true target (y/n)"; B$
80 IF B$ = "Y" OR B$ = "y" THEN T$ = "yes"
90 INPUT "search area radius"; R
100 INPUT "initial range from center"; P
110 IF P > R THEN 100
120 INPUT "search speed"; VS
```

```
130 IF T$ = "no" THEN 170
140 INPUT "true target detection range"; DT
150 IF DT = 0 AND N = 0 THEN 140
160 IF N = 0 THEN 200
170 INPUT "false target detection range"; DF
180 IF DF = 0 AND T$ = "no" THEN 170
190 IF T$ = "no" THEN 240
200 INPUT "classification speed"; VC
210 INPUT "true target classification range"; CT
220 IF CT > DT THEN 210
230 IF N = 0 THEN 260
240 INPUT "false target classification range"; CF
250 IF CF > DF THEN 240
260 INPUT "randomize (y/n)"; D$
270 IF D$ = "Y" OR D$ = "y" THEN R$ = "yes" ELSE 300
280 INPUT "random number seed"; SI
290 IF SI < -32768! OR SI > 32767 THEN 280 ELSE RANDOMIZE SI
300 INPUT "generate a search statistics data file (y/n)"; E$
310 IF E$ = "N" OR E$ = "n" THEN 370
320 INPUT "search statistics data file name"; S$
330 INPUT "generate a search duration histogram data file (y/n)"; F$
340 IF F$ = "N" OR F$ = "n" THEN 370
350 INPUT "search duration histogram file name"; H$
360 INPUT "histogram cell size"; DEL
370 INPUT "display an encounter geometry (y/n)"; G$
380 FOR H = 1 TO M
390 K = N: L = 0: T = 0
400 FOR I = 0 TO N: REM generate the target coordinates
410 GOSUB 2690
420 X(I) = 2 * R * (RD - .5): REM the true target x-coordinate is X(0)
430 GOSUB 2690
440 Y(I) = 2 * R * (RD - .5): REM the true target y-coordinate is Y(0)
450 IF X(I) * X(I) + Y(I) * Y(I) > R * R THEN 410
460 NEXT I
470 IF B$ = "N" OR B$ = "n" THEN X(0) = 10 ^ 10: Y(0) = 10 ^ 10
480 GOSUB 2690
490 XN = (1 - 2 * RD) * P: YN = SQR(P * P - XN * XN)
500 GOSUB 2690
510 SN = SGN(1 - 2 * RD): IF SN = 0 THEN 500
520 XN = XN: YN = SN * YN: REM the initial searcher coordinates
530 IF H > 1 OR NOT (G$ = "Y" OR G$ = "y") THEN 610
540 SCREEN 9: SC = 18.3 / 25 * 64 / 35: WINDOW (-SC * R, -R)-(SC * R, R): T1 = .0175
550 IF B$ = "N" OR B$ = "n" THEN 570
560 T1 = .017: LINE (-T1 * R + X(0), -T1 * R + Y(0))-(T1 * R + X(0), T1 * R + Y(0)), , B
570 FOR I = 1 TO N
580 CIRCLE (X(I), Y(I)), .01 * R
590 NEXT I
600 CIRCLE (0, 0), R
610 FOR Z = 0 TO N
620 J2 = 0: REM boundary scatter flag
630 FOR I = 0 TO K: REM determine the target ranges
640 R(I) = SQR((X(I) - XN) * (X(I) - XN) + (Y(I) - YN) * (Y(I) - YN))
650 NEXT I
660 FOR J = 2 TO K: REM give the closest false target index number 1
670 IF R(J) > R(1) THEN 710
680 A = R(1): R(1) = R(J): R(J) = A
690 A = X(1): X(1) = X(J): X(J) = A
700 A = Y(1): Y(1) = Y(J): Y(J) = A
710 NEXT J
720 IF N = 0 THEN 740
730 IF R(1) <= DF THEN 1330
740 IF R(0) <= DT THEN 1440
750 GOSUB 2690
760 IF J2 = 0 THEN G1 = RD * 2 * PI: GOTO 800: REM G1 is the bearing from the searcher of a random point
770 X = 2 * RD - 1
```

```
780 IF ABS(X) < 1 THEN G1 = ATN(X / SQR(1 - X * X)) + G2: GOTO 800: REM G1 is the boundary scatter angle
790 IF X = 1 THEN G1 = PI / 2 + G2 ELSE G1 = -PI / 2 + G2
800 D = R: REM distance of the random point from the searcher
810 X1 = D * SIN(G1): Y1 = D * COS(G1): REM coordinates of the random point relative to the searcher
820 B1 = X1: IF B1 = 0 THEN 760
830 A1 = -Y1: C1 = Y1 * XN - X1 * YN
840 A2 = -A1 / B1: B2 = -C1 / B1: REM slope and intercept of a target track that passes through the random
point
850 J1 = 0: J2 = 0: D(0) = DT: D(1) = DF
860 FOR I = 0 TO K
870 IF I = 0 THEN J = 0 ELSE J = 1
880 D1(I) = ABS((A1 * X(I) + B1 * Y(I) + C1) / SQR(A1 * A1 + B1 * B1)): REM D1(I) is the distance of object
I from the search track
890 IF D1(I) > D(J) THEN X1(I) = 0: Y1(I) = 0: D1(I) = 2 * R: D2(I) = 2 * R: D3(I) = 2 * R: GOTO 960: REM
900 X1(I) = (X(I) + A2 * Y(I) - A2 * B2) / (1 + A2 * A2)
910 Y1(I) = A2 * X1(I) + B2: REM X1(I) and Y1(I) are the search track CPA coordinates for object I
920 D2(I) = SQR((X1(I) - XN) * (X1(I) - XN) + (Y1(I) - YN) * (Y1(I) - YN)): REM the distance from the
searcher to the detection point for target I
930 D3(I) = SQR(D(J) * D(J) - D1(I) * D1(I)): REM D3(I) is the distance from the CPA to the detection point
for target I
940 D2(I) = D2(I) - D3(I): REM D2(I) is the distance from the searcher to the detection point for target I
950 J1 = 1: REM detection flag
960 NEXT I
970 IF J1 = 0 THEN 1170: REM no detection on the search track leg
980 IF N = 0 THEN J = 0: GOTO 1070
990 FOR J = 2 TO K
1000 IF D2(J) > D2(1) THEN 1050
1010 D2(1) = D2(J)
1020 D3(1) = D3(J)
1030 X1(1) = X1(J)
1040 Y1(1) = Y1(J)
1050 NEXT J
1060 IF D2(0) > D2(1) THEN J = 1 ELSE J = 0
1070 X = X1(J) - XN: Y = Y1(J) - YN: REM coordinates of the nearest CPA when the origin is at XN and YN
1080 GOSUB 2630: REM convert X and Y to polar coordinates
1090 X = D3(J) * COS(N1): Y = D3(J) * SIN(N1): REM coordinates of the nearest CPA when the origin is at the
detection point
1100 X = X1(J) - X: Y = Y1(J) - Y: REM coordinates of the detection point
1110 IF X * X + Y * Y >= R * R THEN 1170: REM the detection point is outside the bounding circle
1120 IF H > 1 OR NOT (G$ = "Y" OR G$ = "y") THEN 1140
1130 LINE (XN, YN)-(X, Y)
1140 GOSUB 2700: REM determine the time to the detection point
1150 XN = X: YN = Y: REM move the searcher to the detection point
1160 GOTO 630: REM prepare to classify
1170 R1 = (A2 * A2 * B2 * B2 - (1 + A2 * A2) * (B2 * B2 - R * R))
1180 IF R1 < 0 THEN R1 = 0: GOTO 1220
1190 R1 = 2 * SQR(R1)
1200 GOSUB 2690
1210 IF .5 - RD > 0 THEN R1 = -R1: REM random choice of one of the two intersection points of the search
track with the bounding circle
1220 GOSUB 2730: REM determine the coordinates of the intersection of the search track with the bounding
circle
1230 IF ABS(X - XN) < 1E-10 * R AND ABS(Y - YN < 1E-10 * R) THEN 1240 ELSE 1260
1240 R1 = -R1
1250 GOSUB 2730
1260 IF H > 1 OR NOT (G$ = "Y" OR G$ = "y") THEN 1280
1270 LINE (XN, YN)-(X, Y)
1280 GOSUB 2700: REM determine the time on the search track to the bounding circle
1290 XN = X: YN = Y: REM move the searcher to the bounding circle
1300 X = -X: Y = -Y: J2 = 1
1310 GOSUB 2630: REM determine the reciprocal direction of the search track
1320 GOTO 630
1330 IF R(0) <= R(1) AND R(0) <= DT THEN 1440: REM the true target is as close or closer than the closest
false target and the true target is detected
1340 IF R(1) <= CF THEN 1370: REM the closest false target is within classification range
```

```
1350 T = T + (R(1) - CF) / VS: REM determine the time from the detection point to the classification point
1360 GOSUB 2470: REM determine the coordinates of the classification point and move the searcher to the
point
1370 L = L + 1
1380 IF B$ = "Y" OR B$ = "y" THEN 1400
1390 IF L = N THEN 1480
1400 GOSUB 2390: REM remove the detected false target from the search
1410 IF B$ = "N" OR B$ = "n" THEN 1430
1420 IF L = N THEN X(1) = 10 ^ 10: Y(1) = 10 ^ 10: R(1) = 10 ^ 10
1430 NEXT Z
1440 IF R(0) <= CT THEN 1470
1450 T = T + (R(0) - CT) / VC: REM determine the time from the detection point to the classification point
1460 GOSUB 2550: REM determine the coordinates of the classification point and move the searcher to the
point
1470 L = L + 1
1480 U = U + T: Q = Q + L
1490 S = T * T + S: W = L * L + W
1500 IF F$ = "N" OR F$ = "n" THEN 1550
1510 F = INT(T)
1520 IF F < 7000 THEN 1540
1530 H(7000) = H(7000) + 1: GOTO 1550
1540 H(F) = H(F) + 1
1550 NEXT H
1560 SCREEN 0
1570 U = U / M: Q = Q / M
1580 IF M > 1 THEN 1600
1590 S = 0: W = 0: GOTO 1610
1600 S = SQR(S / M - U * U): W = SQR(W / M - Q * Q)
1610 PRINT : PRINT : PRINT "average search duration ="; U
1620 PRINT "standard deviation ="; S
1630 PRINT "average number of classifications ="; Q
1640 PRINT "standard deviation ="; W
1650 IF E$ = "N" OR E$ = "n" THEN 1850
1660 OPEN "O", #1, S$
1670 WRITE #1, M, N, R, P, VS, DT, DF, VC, CT, CF, U, S, Q, W, F$, S$, R$, T$
1680 IF F$ = "N" OR F$ = "n" THEN CLOSE : GOTO 1850
1690 FOR I = 0 TO 7000
1700 IF H(I) > 0 THEN J = I
1710 NEXT I
1720 L = J / DEL: L = INT(L)
1730 DIM K(L + 1)
1740 FOR I = 0 TO J
1750 G = H(I): F = I / DEL: F = INT(F): K(F) = K(F) + G
1760 NEXT I
1770 WRITE #1, H$, DEL, L, J
1780 CLOSE
1790 OPEN "O", #1, H$
1800 FOR I = 0 TO L
1810 WRITE #1, K(I)
1820 NEXT I
1830 CLOSE
1840 PRINT "histogram file entries = " + STR$(L + 1)
1850 PRINT : PRINT : PRINT : INPUT "print data (y/n)"; A$
1860 IF A$ = "N" OR A$ = "n" THEN 2380
1870 LPRINT : LPRINT : LPRINT "program name = TARGET.BAS"
1880 LPRINT "number of trials ="; M
1890 LPRINT "number of false targets ="; N
1900 LPRINT "true target: " + T$
1910 LPRINT "search area radius ="; R
1920 LPRINT "initial range from center ="; P
1930 LPRINT "search speed ="; VS
1940 IF T$ = "no" THEN 1970
1950 LPRINT "true target detection range ="; DT
1960 IF N = 0 THEN 1980
1970 LPRINT "false target detection range ="; DF
```

```
1980 LPRINT "classification speed ="; VC
1990 IF T$ = "no" THEN 2020
2000 LPRINT "true target classification range ="; CT
2010 IF N = 0 THEN 2030
2020 LPRINT "false target classification range ="; CF
2030 LPRINT "randomize: " + R$
2040 IF E$ = "N" OR E$ = "n" THEN 2090
2050 LPRINT "search statistics data file name = " + S$
2060 IF F$ = "N" OR F$ = "n" THEN 2090
2070 LPRINT "search duration histogram data file name = " + H$
2080 LPRINT "histogram cell size = " + STR$(DEL)
2090 LPRINT : LPRINT : LPRINT "average search duration ="; U
2100 LPRINT "standard deviation ="; S
2110 LPRINT "average number of classifications ="; Q
2120 LPRINT "standard deviation ="; W
2130 IF F$ = "N" OR F$ = "n" THEN 2380
2140 LPRINT "histogram file entries = " + STR$(L + 1)
2150 IF J < 7000 THEN 2170
2160 LPRINT "file truncated"
2170 PRINT : INPUT "print histogram file entries (y/n)"; G$
2180 IF G$ = "N" OR G$ = "n" THEN 2270
2190 LPRINT : LPRINT : LPRINT
2200 LPRINT H$ + " histogram file entries"
2210 LPRINT
2220 LPRINT "time"; TAB(15); "number"
2230 LPRINT
2240 FOR I = 0 TO L
2250 LPRINT I, K(I)
2260 NEXT I
2270 PRINT : INPUT "print cumulative histogram file entries (y/n)"; K$
2280 IF K$ = "N" OR K$ = "n" THEN 2380
2290 LPRINT : LPRINT : LPRINT
2300 LPRINT H$ + " cumulative histogram file entries"
2310 LPRINT
2320 LPRINT "time"; TAB(15); "number"
2330 LPRINT
2340 FOR I = 0 TO L
2350 KSUM = K(I) + KSUM
2360 LPRINT I, KSUM
2370 NEXT I
2380 END
2390 IF K = 1 THEN X(1) = 2 * R: Y(1) = 2 * R: GOTO 2460
2400 FOR I = 1 TO K - 1
2410 X(I) = X(I + 1)
2420 Y(I) = Y(I + 1)
2430 R(I) = R(I + 1)
2440 NEXT I
2450 K = K - 1
2460 RETURN
2470 X = XN - X(1): Y = YN - Y(1)
2480 GOSUB 2630
2490 X = CF * COS(N1): Y = CF * SIN(N1)
2500 X = X + X(1): Y = Y + Y(1)
2510 IF H > 1 OR NOT (G$ = "Y" OR G$ = "y") THEN 2530
2520 LINE (XN, YN)-(X, Y)
2530 XN = X: YN = Y
2540 RETURN
2550 X = XN - X(0): Y = YN - Y(0)
2560 GOSUB 2630
2570 X = CT * COS(N1): Y = CT * SIN(N1)
2580 X = X + X(0): Y = Y + Y(0)
2590 IF H > 1 OR NOT (G$ = "Y" OR G$ = "y") THEN 2610
2600 LINE (XN, YN)-(X, Y)
2610 XN = X: YN = Y
2620 RETURN
```

```
2630 R1 = SQR(X * X + Y * Y)
2640 IF ABS(X / R1) = 1 THEN M1 = PI / 2 * SGN(X) ELSE M1 = ATN(X / R1 / SQR(1 - X * X / R1 / R1))
2650 IF ABS(Y / R1) = 1 THEN N1 = PI / 2 * (1 - SGN(Y)) ELSE N1 = PI / 2 - ATN(Y / R1 / SQR(1 - Y * Y / R1 /
R1))
2660 IF M1 < 0 THEN N1 = 2 * PI - N1
2670 IF J2 = 1 THEN G2 = N1: REM circle radial direction
2680 N1 = PI / 2 - N1: RETURN
2690 RD = RND: RETURN: REM location for an alternate random number generator
2700 R5 = SQR((X - XN) * (X - XN) + (Y - YN) * (Y - YN))
2710 T = T + R5 / VS
2720 RETURN
2730 X = (-2 * A2 * B2 + R1) / (2 * (1 + A2 * A2))
2740 Y = A2 * X + B2
2750 RETURN
```

Appendix 2.

The program listed below is designed to read and then print the data files that are generated by TARGET.BAS.

```
10 REM TARGETPR.BAS  a program to print TARGET.BAS data files
20 CLS
30 ON ERROR GOTO 40: GOTO 50
40 RESUME 20
50 INPUT "search statistics data file name"; S$
60 OPEN "I", #1, S$
70 INPUT #1, M, N, R, P, VS, DT, DF, VC, CT, CF, U, S, Q, W, F$, S$, R$, T$
80 IF F$ = "N" OR F$ = "n" THEN 100
90 INPUT #1, H$, DEL, L, J
100 CLOSE
110 LPRINT : LPRINT : LPRINT "program name = TARGET.BAS"
120 LPRINT "number of trials ="; M
130 LPRINT "number of false targets ="; N
140 LPRINT "true target: " + T$
150 LPRINT "search area radius ="; R
160 LPRINT "initial range from center ="; P
170 LPRINT "search speed ="; VS
180 IF T$ = "no" THEN 210
190 LPRINT "true target detection range ="; DT
200 IF N = 0 THEN 220
210 LPRINT "false target detection range ="; DF
220 LPRINT "classification speed ="; VC
230 IF T$ = "no" THEN 260
240 LPRINT "true target classification range ="; CT
250 IF N = 0 THEN 270
260 LPRINT "false target classification range ="; CF
270 LPRINT "randomize: " + R$
280 LPRINT "search statistics data file name = " + S$
290 IF F$ = "N" OR F$ = "n" THEN 320
300 LPRINT "search duration histogram data file name = " + H$
310 LPRINT "histogram cell size = " + STR$(DEL)
320 LPRINT : LPRINT : LPRINT "average search duration ="; U
330 LPRINT "standard deviation ="; S
340 LPRINT "average number of classifications ="; Q
350 LPRINT "standard deviation ="; W
360 IF F$ = "N" OR F$ = "n" THEN 670
370 LPRINT "histogram file entries = " + STR$(L + 1)
380 IF J < 7000 THEN 400
390 LPRINT "file truncated"
400 PRINT : INPUT "print histogram file entries (y/n)"; A$
410 IF A$ = "N" OR A$ = "n" THEN 560
420 OPEN "I", #1, H$
430 DIM K(L)
440 FOR I = 0 TO L
450 INPUT #1, K(I)
460 NEXT I
470 CLOSE
480 LPRINT : LPRINT : LPRINT
490 LPRINT H$ + " histogram file entries"
500 LPRINT
510 LPRINT "time"; TAB(15); "number"
520 LPRINT
530 FOR I = 0 TO L
540 LPRINT I, K(I)
550 NEXT I
560 PRINT : INPUT "print cumulative histogram file entries (y/n)"; K$
```

```
570 IF K$ = "N" OR K$ = "n" THEN 670
580 LPRINT : LPRINT : LPRINT
590 LPRINT H$ + " cumulative histogram file entries"
600 LPRINT
610 LPRINT "time"; TAB(15); "number"
620 LPRINT
630 FOR I = 0 TO L
640 KSUM = K(I) + KSUM
650 LPRINT I, KSUM
660 NEXT I
670 END
```

Appendix 3.

The random search model, for a search at a constant speed v in an search region of area A by a searcher with a detection system has a sweep width W for a target, gives the following for the probability that the random variable T, the duration of the search for a target that is in the region will be less than or equal to t:

$$P(T \leq t) = 1 - \exp(-W \cdot v \cdot t/A)$$

As a consequence of this, $t_{0.5}$, the median search time, is given by the following expression:

$$t_{0.5} = -[A/(W \cdot v)] \cdot \ln(0.5)$$

For the simulation search model, the sweep width is equal to twice the detection range.

Appendix 4.

Figure 7 and Figure 8 illustrate the screen display when the option "display an encounter geometry" is chosen. The encounter geometry that is displayed when the option is chosen is the encounter geometry of the first trial simulation. The display is maintained until the simulation trials are complete or the program is terminated by a break.

Figures are shown in Reference 1 that were generated by Chudnovsky and Chudnovsky by presumably using the same reflection law that is used in the simulation program that is listed in Appendix 1 and that generated Figure 7 and Figure 8.



Figure 7 The target is located at the center of the square. The false targets are located at the centers of the 15 rectangles. The radius is 50 nautical miles, the detection ranges are 15 nautical miles and the classification ranges are 5 nautical miles.



Figure 8 The target is located at the center of the square. There are no false targets. The radius is 100 nautical miles, the detection range is 1 nautical mile and the classification range is 1 nautical mile.

References

1. Chudnovsky, D. V. and Chudnovsky, G. V., Search Theory, Marcel Dekker, Inc., New York, 1989.

2. Haley, K. B. and Stone, L. D., Search Theory and Applications, Plenum Press, New york, 1980.

3. Washburn, A. R., Search and Detection, Military Applications Section, Operations Research Society of America, May 1981.

Initial Distribution List

	Copies
Director of Research Administration (Code 012) Naval Postgraduate School Monterey, CA 93943-5000	1
Defense Technical Information Center Cameron Station Alexandria, VA 22314	2
Library (Code 0142) Naval Postgraduate School Monterey, CA 93943-5000	2
Chief of Naval Operations OP-714C Washington, D. C. 20350-2000	1
Commander Submarine Development Squadron 12 Naval Submarine Base, New London Groton, CT 06349	1
Commander Surface Warfare Development Group Naval Amphibious Base, Little Creek Norfolk, VA 23521	1
Naval Air Development Center Johnsville, PA 18974	1
Naval Surface Weapons Center White Oak Silver Spring, MD 20910	1
Naval Underwater Systems Center Newport, RI 02840	1
Naval Underwater Systems Center New London, CT 06320	1
Naval Technical Intelligence Center 4301 Suitland Road Washington, D. C. 20390	1
Naval Research Laboratory Washington, D. C. 20375	1
Naval Ocean Systems Center San Diego, CA 92132	1

Center for Naval Analysis 1
4401 Ford Avenue
P. O. Box 16268
Alexandria, VA 22302-0268

Naval Postgraduate School 50
Monterey, CA 93943-5000
Code 55Fo